4.5 Earth to the Moon

A projectile launched from the surface of the earth is attracted both by the earth and the moon. The altitude $r(t)$ of the projectile above the earth is known to satisfy the initial value problem (see *Technical Details* page 261)

$$\begin{aligned} r''(t) &= -\frac{Gm_1}{(R_1 + r(t))^2} + \frac{Gm_2}{(R_2 - R_1 - r(t))^2}, \\ r(0) &= 0, \quad r'(0) = v_0. \end{aligned} \tag{1}$$

The unknown initial velocity v_0 of the projectile is given in meters per second. The constants in (1) are defined as follows.

$G = 6.6726 \times 10^{-11}$ N-m^2/kg^2	Universal gravitation constant,
$m_1 = 5.975 \times 10^{24}$ kilograms	Mass of the earth,
$m_2 = 7.36 \times 10^{22}$ kilograms	Mass of the moon,
$R_1 = 6,378,000$ meters	Radius of the earth,
$R_2 = 384,400,000$ meters	Distance from the earth's center to the moon's center.

The Jules Verne Problem

In his 1865 novel *From the Earth to the Moon*, Jules Verne asked what initial velocity must be given to the projectile in order to reach the moon. The question in terms of equation (1) becomes:

> What minimal value of v_0 causes the projectile to have zero net acceleration at some point between the earth and the moon?

The projectile only has to travel a distance R equal to the surface-to-surface distance between the earth and the moon. The altitude $r(t)$ of the projectile must satisfy $0 \le r \le R$. Given v_0 for which the net acceleration is zero, $r''(t) = 0$ in (1), then the projectile has reached a critical altitude r^*, where gravitational effects of the moon take over and the projectile will fall to the surface of the moon.

Let $r''(t) = 0$ in (1) and substitute r^* for $r(t)$ in the resulting equation. Then

$$\begin{aligned} -\frac{Gm_1}{(R_1 + r^*)^2} + \frac{Gm_2}{(R_2 - R_1 - r^*)^2} &= 0, \\ r^* = \frac{R_2}{1 + \sqrt{m_2/m_1}} - R_1 &\approx 339,620,820 \text{ meters}. \end{aligned} \tag{2}$$

Using energy methods (see *Technical details*, page 262), it is possible to calculate exactly the *minimal* earth-to-moon velocity v_0^* required for the projectile to *just reach* critical altitude r^*:

(3) $$v_0^* \approx 11067.31016 \quad \text{meters per second.}$$

A Numerical Experiment

The value $v_0^* \approx 11067.31016$ in (3) will be verified experimentally. As part of this experiment, the flight time is estimated.

Such a numerical experiment must adjust the initial velocity v_0 in initial value problem (1) so that $r(t)$ increases from 0 to R. Graphical analysis of a solution $r(t)$ for low velocities v_0 gives insight into the problem; see Figure 7.

The choice of numerical software solver makes for significant differences in this problem. Initial work used the Livermore Laboratory numerical stiff solver for ordinary differential equations (acronym `lsode`).

Computer algebra system `maple` documents and implements algorithm `lsode` with `dsolve` options of `method=lsode` or `stiff=true`. Other stiff solvers of equal quality can be used for nearly identical results. Experiments are necessary to determine if the required accuracy has been attained.



Figure 7. Jules Verne Problem. The solution $r(t)$ of (1) for $v_0 = 1000$. The projectile rises to a maximum height of about $51,427$ meters, then it falls back to earth. The trip time is 206 seconds.

The numerical experiment solves (1) using `lsode`, then the solution is graphed, to see if the projectile falls back to earth (as in Figure 7) or if it reaches an altitude near r^* and then falls to the moon. Suitable starting values for the initial velocity v_0 and the trip time T are $v_0 = 1000$ and $T = 210$ (see Figure 7), in the case when the projectile falls back to earth. The projectile travels to the moon when the r-axis of the graphic has maximum greater than $r^* \approx 339,620,820$ meters. The logic is that this condition causes the gravitation effects of the moon to be strong enough to force the projectile to fall to the moon.

In Table 20 appears `maple` initialization code. In Table 21, group 2 is executed a number of times, to refine estimates for the initial velocity v_0 and the trip time T. A summary of some estimates appear in Table 22. The graphics produced along the way resemble Figure 7 or Figure 8. A successful trip to the moon is represented in Figure 8, which uses $v_0 = 11068$ meters per second and $T = 527000$ seconds.



Figure 8. Experimental trip to the moon. The initial velocity is $v_0 = 24,764$ miles per hour and the trip time is 147 hours. See Table 22 for details about how these values were obtained.

Table 20. Initialization code in `maple` for the numerical experiment. Group 1 defines seven constants G, m_1, m_2, R_1, R_2, R_3, R and computes values $r^* \approx 339,620,820$ and $v_0^* \approx 11067.31016$.

```
# Group 1:  Constants plus rstar and v0star
G:=6.6726e-11:  m1:=5.975e24:  m2:=7.36e22:
R1:=6.378e6:  R2:=3.844e8:  R3:=1.74e6:
R:=R2-R1-R3:
ans:=[solve(-G*m1/(r+R1)^2 + G*m2/(R2-R1-r)^2=0,r)]:
rstar:=ans[1];
FF:=r->G*m1/(R1+r)+G*m2/(R2-R1-r):
v0star:=sqrt(2*(FF(0)-FF(rstar)));
```

Table 21. Iteration code in `maple` for the numerical experiment. Group 2 plots a graphic for given v_0 and T. A successful trip to the moon must use velocity $v_0 > v_0^* \approx 11067.31016$. The relation $\max_{0\le t\le T} Y(t) > r^* \approx 339,620,820$ must be valid. Finally, $Y(T) \ge R$ must hold.

```
# Group 2:  Iteration code
v0:=1000:  # v0<v0star.  Projectile falls to earth.
de:=diff(r(t),t,t)=-G*m1/(r(t)+R1)^2+G*m2/(R2-R1-r(t))^2:
ic:=r(0)=0,D(r)(0)=v0:
p:=dsolve({de,ic},r(t),
type=numeric,method=lsode,startinit=true);
Y:=t->rhs(p(t)[2]):
T:=210:  # Guess the trip time T
plot('Y(t)',t=0..T);
plot(['Y(t)',R,rstar],t=0..T); # Add two horizontal lines
# Plot done.  Change v0, T and re-execute group 2.
```

Table 22. Experimental results with the `lsode` solver to obtain estimates for the initial velocity v_0 and the trip time T.

v_0	T	Results
11000	38500	$r(t) < r^*$ for all t
12000	80000	$r(T) > r^*$
11125	200000	$r(T) > r^*$
11060	780000	$r(t) < r^*$ for all t
11070	377500	$r(T) > r^*$
11068	527000	$r(T) \approx R$

Exact trip time. The time T for a trip with velocity $v_0 = 11068$ can be computed once an approximate value for the trip time is known. For instance, if $T = 527000$ gives a successful plot, but $T = 525000$ does not, then the exact value of T is between 525000 and 527000. The computer algebra system can be used to determine the more precise value $T = 527440.9891$, as follows.

```
# Group 2
v0:=11068:  # Projectile reaches the moon.
de:=diff(r(t),t,t)=-G*m1/(r(t)+R1)^2
+G*m2/(R2-R1-r(t))^2:
ic:=r(0)=0,D(r)(0)=v0:
p:=dsolve({de,ic},r(t),
type=numeric,method=lsode,startinit=true);
Y:=t->rhs(p(t)[2]):
fsolve('Y(t)'=R,t=526000);
# T==5.274409891*10^5
```

Details for (1) and (3)

Technical details for (1): To derive (1), it suffices to write down a competition between the Newton's second law force relation $mr''(t)$ and the sum of two forces due to gravitational attraction for the earth and the moon. Here, m stands for the mass of the projectile.

Gravitational force for the earth. This force, by Newton's universal gravitation law, has magnitude

$$F_1 = \frac{Gm_1 m}{\mathcal{R}_3^2}$$

where m_1 is the mass of the earth, G is the universal gravitation constant and $\mathcal{R}_3$ is the distance from the projectile to the center of the earth: $\mathcal{R}_3 = R_1 + r(t)$.

Gravitational force for the moon. Similarly, this force has magnitude

$$F_2 = \frac{Gm_2 m}{\mathcal{R}_4^2}$$

where m_2 is the mass of the moon and $\mathcal{R}_4$ is the distance from the projectile to the center of the moon: $\mathcal{R}_4 = R_2 - R_1 - r(t)$.

Competition between forces. The force equation is

$$mr''(t) = -F_1 + F_2$$

due to the directions of the force vectors. Simplifying the relations and cancelling m gives equation (1).

Technical details for (3): To justify the value for v_0, multiply equation (1) by r' and integrate the new equation from $t = 0$ to $t = t_0$ to get

$$(4)\qquad \begin{aligned} &\frac{1}{2}(r'(t_0))^2 = F(r(t_0)) - F(0) + \frac{1}{2}v_0^2, \quad \text{where} \\ &F(r) = \frac{Gm_1}{R_1 + r} + \frac{Gm_2}{R_2 - R_1 - r}. \end{aligned}$$

The expression $F(r)$ is minimized when $F'(r) = 0$ or else at $r = 0$ or $r = R$. The right side of (1) is $F'(r)$, hence $F(r)$ has unique critical point $r = r^*$. Compute $F(0) = 62522859.35$, $F(r^*) = 1280168.523$ and $F(R) = 3864318.458$. Then the minimum of $F(r)$ is at $r = r^*$ and $F(r^*) \le F(r(t_0))$.

The left side of the first equality in (4) is nonnegative, therefore also the right side is nonnegative, giving $\frac{1}{2}v_0^2 \ge F(0) - F(r(t_0))$. If the projectile ever reaches altitude r^*, then $r(t_0) = r^*$ is allowed and $v_0 \ge \sqrt{2F(0) - 2F(r^*)} \approx 11067.31016$. Restated, $v_0 < 11067.31016$ implies the projectile *never reaches altitude* r^*, hence it falls back to earth. On the other hand, if $v_0 > 11067.31016$, then by (4) and $F(r^*) \le F(r)$ it follows that $r'(t) > 0$ and therefore the projectile cannot return to earth. That is, $r(t) = 0$ for some $t > 0$ can't happen.

In summary, the least launch velocity v_0^* which allows $r(t) = r^*$ for some $t > 0$ is given by the formulas

$$v_0^* = \sqrt{2F(0) - 2F(r^*)}, \quad F(r) = \frac{Gm_1}{R_1 + r} + \frac{Gm_2}{R_2 - R_1 - r}.$$

This completes the proof of equation (3).

Exercises 4.5

Critical Altitude r^*. The symbol r^* is the altitude $r(t)$ at which gravitational effects of the moon take over, causing the projectile to fall to the moon.

1. Justify from the differential equation that $r''(t) = 0$ at $r^* = r(t)$ implies the first relation in (2):

$$\frac{Gm_2}{(R_2 - R_1 - r^*)^2} - \frac{Gm_1}{(R_1 + r^*)^2} = 0.$$

2. Solve symbolically the relation of the previous exercise for r^*, to obtain the second equation of (2):

$$r^* = \frac{R_2}{1 + \sqrt{m_2/m_1}} - R_1.$$

3. Use the previous exercise and values for the constants R_1, R_2, m_1, m_2 to obtain the approximation

$$r^* = 339,620,820 \text{ meters}.$$

4. Determine the effect on r^* for a one percent error in measurement m_2. Replace m_2 by $0.99m_2$ and $1.01m_2$ in the formula for r^* and report the two estimated critical altitudes.

Escape Velocity v_0^*. The symbol v_0^* is the velocity $r'(0)$ such that $\lim_{t\to\infty} r(t) = \infty$, but smaller launch velocities will cause the projectile to fall back to the earth. Throughout, define

$$F(r) = \frac{Gm_1}{R_1 + r} + \frac{Gm_2}{R_2 - R_1 - r}.$$

5. Let $v_0 = r'(0)$, $r^* = r(t_0)$. Derive the formula

$$\frac{1}{2}(r'(t_0))^2 = F(r^*) - F(0) + \frac{1}{2}v_0^2$$

which appears in the proof details.

6. Verify using the previous exercise that $r'(t_0) = 0$ implies

$$v_0^* = \sqrt{2(F(0) - F(r^*))}.$$

7. Verify by hand calculation that $v_0^* \approx 11067.31016$ meters per second.

8. Argue by mathematical proof that $F(r)$ is not minimized at the endpoints of the interval $0 \le r \le R$.

Numerical Experiments. Assume values given in the text for physical constants. Perform the given experiment, using numerical software, on initial value problem (1), page 258. The cases when $v_0 > v_0^*$ escape the earth, while the others fall back to earth.

9. RK4 solver, $v_0 = 11068$, $T = 515000$. Plot the solution on $0 \le t \le T$.

10. Stiff solver, $v_0 = 11068$, $T = 515000$. Plot the solution on $0 \le t \le T$.

11. RK4 solver, $v_0 = 11067.2$, $T = 800000$. Plot the solution on $0 \le t \le T$.

12. Stiff solver, $v_0 = 11067.2$, $T = 800000$. Plot the solution on $0 \le t \le T$.

13. RK4 solver, $v_0 = 11067$, $T = 1000000$. Plot the solution on $0 \le t \le T$.

14. Stiff solver, $v_0 = 11067$, $T = 1000000$. Plot the solution on $0 \le t \le T$.

15. RK4 solver, $v_0 = 11066$, $T = 800000$. Plot the solution on $0 \le t \le T$.

16. Stiff solver, $v_0 = 11066$, $T = 800000$. Plot the solution on $0 \le t \le T$.

17. RK4 solver, $v_0 = 11065$. Find a suitable value T which shows that the projectile falls back to earth, then plot the solution on $0 \le t \le T$.

18. Stiff solver, $v_0 = 11065$. Find a suitable value T which shows that the projectile falls back to earth, then plot the solution on $0 \le t \le T$.

19. RK4 solver, $v_0 = 11070$. Find a suitable value T which shows that the projectile falls to the moon, then plot the solution on $0 \le t \le T$.

20. Stiff solver, $v_0 = 11070$. Find a suitable value T which shows that the projectile falls to the moon, then plot the solution on $0 \le t \le T$.

4.6 Skydiving

A skydiver of 160 pounds jumps from a hovercraft at 15,000 feet. The fall is mostly vertical from zero initial velocity, but there are significant effects from air resistance until the parachute opens at 5,000 feet. The resistance effects are determined by the skydiver's clothing and body shape.

Velocity Model. Assume the skydiver's air resistance is modeled in terms of velocity v by a force equation

$$F(v) = av + bv^2 + cv^3.$$

The constants a, b, c are given by the formulas

$$a = 0.009, \quad b = 0.0008, \quad c = 0.0001.$$

In particular, the force $F(v)$ is positive for v positive. According to Newton's second law, the velocity $v(t)$ of the skydiver satisfies $mv'(t) = mg - F(v)$. We assume $mg = 160$ pounds and $g \approx 32$ feet per second per second. The **velocity model** is

$$v'(t) = 32 - \frac{32}{160}\left(0.009v(t) + 0.0008v^2(t) + 0.0001v^3(t)\right), \quad v(0) = 0.$$

Distance Model. The distance $x(t)$ traveled by the skydiver, measured from the hovercraft, is given by the **distance model**

$$x'(t) = v(t), \quad x(0) = 0.$$

The velocity is expected to be positive throughout the flight. Because the parachute opens at 5000 feet, at which time the velocity model must be replaced the open parachute model (not discussed here), the distance $x(t)$ increases with time from 0 feet to its limiting value of 10000 feet. Values of $x(t)$ from 10000 to 15000 feet make sense only for the open parachute model.

Terminal Velocity. The **terminal velocity** is an equilibrium solution $v(t) = v_\infty$ of the velocity model, therefore constant v_∞ satisfies

$$32 - \frac{32}{160}\left(0.009v_\infty + 0.0008v_\infty^2 + 0.0001v_\infty^3\right) = 0.$$

A numerical solver is applied to find the value $v_\infty = 114.1$ feet per second, which is about 77.8 miles per hour. For the solver, we define $f(v) = 32 - F(v)$ and solve $f(v) = 0$ for v. Some `maple` details:

```
f:=v->32 - (32/160)*(0.009*v+0.0008*v^2+0.0001*v^3);
fsolve(f(v)=0,v);              # 114.1032777 ft/sec
60*60*fsolve(f(v)=0,v)/5280;  # 77.79768934 mi/hr
```

A Numerical Experiment. The Runge-Kutta method will be applied to produce a table which contains the elapsed time t, the skydiver velocity $v(t)$ and the distance traveled $x(t)$, up until the distance reaches nearly 10000 feet, whereupon the parachute opens.

The objective here is to illustrate practical methods of table production in a computer algebra system or numerical laboratory. It is efficient in these computational systems to phrase the problem as a system of two differential equations with two initial conditions.

System Conversion. The velocity substitution $v(t) = x'(t)$ used in the velocity model gives us two differential equations in the unknowns $x(t)$, $v(t)$:

$$x'(t) = v(t),\ v'(t) = g - \frac{1}{m}F(v(t)).$$

Define $f(v) = g - (1/m)F(v)$. The path we follow is to execute the `maple` code below, which produces the table that follows using the default Runge-Kutta-Fehlberg algorithm.

```
eq:=32 - (32/160)*(0.009*v+0.0008*v^2+0.0001*v^3:
f:=unapply(eq,v);
de1:=diff(x(t),t)=v(t); de2:=diff(v(t),t)=f(v(t));
ic:=x(0)=0,v(0)=0;opts:=numeric,output=listprocedure:
p:=dsolve({de1,de2,ic},[x(t),v(t)],opts);
X:=eval(x(t),p); V:=eval(v(t),p);
fmt:="%10.2f  %10.2f  %10.2f\n";
seq(printf(fmt,5*t,X(5*t),V(5*t)),t=0..18);
```

t	$x(t)$	$v(t)$	t	$x(t)$	$v(t)$
5.00	331.26	106.84	50.00	5456.76	114.10
10.00	892.79	113.97	55.00	6027.28	114.10
15.00	1463.15	114.10	60.00	6597.80	114.10
20.00	2033.67	114.10	65.00	7168.31	114.10
25.00	2604.18	114.10	70.00	7738.83	114.10
30.00	3174.70	114.10	75.00	8309.35	114.10
35.00	3745.21	114.10	80.00	8879.86	114.10
40.00	4315.73	114.10	85.00	9450.38	114.10
45.00	4886.25	114.10	90.00	10020.90	114.10

The table says that the flight time to parachute open at 10,000 feet is about 90 seconds and the terminal velocity 114.10 feet/sec is reached in about 15 seconds.

More accurate values for the flight time 89.82 to 10,000 feet and time 14.47 to terminal velocity can be determined as follows.

```
fsolve(X(t)=10000,t,80..95);
fsolve(V(t)=114.10,t,2..20);
```

Alternate Method. Another way produce the table is to solve the velocity model numerically, then determine $x(t) = \int_0^t v(r)dr$ by numerical integration. Due to accuracy considerations, a variant of Simpson's rule is used, called the **Newton-cotes rule**. The `maple` implementation of this idea follows.

The first method of conversion into two differential equations is preferred, even though the alternate method reproduces the table using only the textbook material presented in this chapter.

```
f:=unapply(32-(32/160)*(0.009*v+0.0008*v^2+0.0001*v^3),v);
de:=diff(v(t),t)=f(v(t)); ic:=v(0)=0;
q:=dsolve({de,ic},v(t),numeric);
V:=t->rhs(q(t)[2]);
X:=u->evalf(Int(V,0..u,continuous,_NCrule));
fmt:="%10.2f  %10.2f  %10.2f\n";
seq(printf(fmt,5*t,X(5*t),V(5*t)),t=0..18);
```

Ejected Baggage. Much of what has been done here applies as well to an ejected parcel, instead of a skydiver. What changes is the force equation $F(v)$, which depends upon the parcel exterior and shape. The distance model remains the same, but the restraint $0 \le x \le 10000$ no longer applies, since no parachute opens. We expect the parcel to reach terminal velocity in 5 to 10 seconds and hit the ground at that speed.

Variable Mass. The mass of a skydiver can be time-varying. For instance, the skydiver lets water leak from a reservoir. This kind of problem assumes mass $m(t)$, position $x(t)$ and velocity $v(t)$ for the diver. Then Newton's second law gives a position-velocity model

$$\begin{aligned} &x'(t) = v(t), \\ &(m(t)v(t))' = G(t, x(t), v(t)). \end{aligned}$$

The problem is similar to rocket propulsion, in which expended fuel decreases the in-flight mass of the rocket. Simplifying assumptions make it possible to present formulas for $m(t)$ and $G(t, x, v)$, which can be used by the differential equation solver.

Exercises 4.6

Terminal Velocity. Assume force $F(v) = av + bv^2 + cv^3$ and $g = 32$, $m = 160/g$. Using computer assist, find the terminal velocity v_∞ from the velocity model $v' = g - \frac{1}{m}F(v)$, $v(0) = 0$.

1. $a = 0$, $b = 0$ and $c = 0.0002$.

2. $a = 0$, $b = 0$ and $c = 0.00015$.

3. $a = 0$, $b = 0.0007$ and $c = 0.00009$.

4. $a = 0$, $b = 0.0007$ and $c = 0.000095$.

5. $a = 0.009$, $b = 0.0008$ and $c = 0.00015$.

6. $a = 0.009$, $b = 0.00075$ and $c = 0.00015$.

7. $a = 0.009$, $b = 0.0007$ and $c = 0.00009$.

8. $a = 0.009$, $b = 0.00077$ and $c = 0.00009$.

9. $a = 0.009$, $b = 0.0007$ and $c = 0$.

10. $a = 0.009$, $b = 0.00077$ and $c = 0$.

Numerical Experiment. Assume the skydiver problem (**??**) with $g = 32$ and constants m, a, b, c supplied below. Using computer assist, apply a numerical method to produce a table for the elapsed time t, the velocity $v(t)$ and the distance $x(t)$. The table must end at $x(t) \approx 10000$ feet, which determines the flight time.

11. $m = 160/g$, $a = 0$, $b = 0$ and $c = 0.0002$.

12. $m = 160/g$, $a = 0$, $b = 0$ and $c = 0.00015$.

13. $m = 130/g$, $a = 0$, $b = 0.0007$ and $c = 0.00009$.

14. $m = 130/g$, $a = 0$, $b = 0.0007$ and $c = 0.000095$.

15. $m = 180/g$, $a = 0.009$, $b = 0.0008$ and $c = 0.00015$.

16. $m = 180/g$, $a = 0.009$, $b = 0.00075$ and $c = 0.00015$.

17. $m = 170/g$, $a = 0.009$, $b = 0.0007$ and $c = 0.00009$.

18. $m = 170/g$, $a = 0.009$, $b = 0.00077$ and $c = 0.00009$.

19. $m = 200/g$, $a = 0.009$, $b = 0.0007$ and $c = 0$.

20. $m = 200/g$, $a = 0.009$, $b = 0.00077$ and $c = 0$.

Flight Time. Assume the skydiver problem (**??**) with $g = 32$ and constants m, a, b, c supplied below. Using computer assist, apply a numerical method to find accurate values for the flight time to 10,000 feet and the time required to reach terminal velocity.

21. $mg = 160$, $a = 0.0095$, $b = 0.0007$ and $c = 0.000092$.

22. $mg = 160$, $a = 0.0097$, $b = 0.00075$ and $c = 0.000095$.

23. $mg = 240$, $a = 0.0092$, $b = 0.0007$ and $c = 0$.

24. $mg = 240$, $a = 0.0095$, $b = 0.00075$ and $c = 0$.

Ejected Baggage. Baggage of 45 pounds is dropped from a hovercraft at $15,000$ feet. Assume air resistance force $F(v) = av + bv^2 + cv^3$, $g = 32$ and $mg = 45$. Using computer assist, find accurate values for the flight time to the ground and the terminal velocity. Estimate the time required to reach 99.95% of terminal velocity.

25. $a = 0.0095$, $b = 0.0007$, $c = 0.00009$

26. $a = 0.0097$, $b = 0.00075$, $c = 0.00009$

27. $a = 0.0099$, $b = 0.0007$, $c = 0.00009$

28. $a = 0.0099$, $b = 0.00075$, $c = 0.00009$

4.7 Lunar Lander

A lunar lander goes through free fall to the surface of the moon, its descent controlled by retrorockets that provide a constant deceleration to counter the effect of the moon's gravitational field.

The retrorocket control is supposed to produce a **soft touchdown**, which means that the velocity $v(t)$ of the lander is zero when the lander touches the moon's surface. To be determined:

H = height above the moon's surface for retrorocket activation,

T = flight time from retrorocket activation to soft touchdown.

Investigated here are two models for the lunar lander problem. In both cases, it is assumed that the lander has mass m and falls in the direction of the moon's gravity vector. The initial speed of the lander is assumed to be v_0. The retrorockets supply a constant thrust deceleration g_1. Either the fps or mks unit system will be used. Expended fuel ejected from the lander during thrust will be ignored, keeping the lander mass constantly m.

The distance $x(t)$ traveled by the lander t time units after retrorocket activation is given by

$$x(t) = \int_0^t v(r)dr, \quad 0 \le t \le T.$$

Therefore, H and T are related by the formulas

$$v(T) = 0, \quad x(T) = H.$$

Constant Gravitational Field

Let g_0 denote the constant acceleration due to the moon's gravitational field. Assume given initial velocity v_0 and the retrorocket thrust deceleration g_1. Define $A = g_1 - g_0$, the effective thrust. Set the origin of coordinates at the center of mass of the lunar lander. Let vector $\vec{\imath}$ have tail at the origin and direction towards the center of the moon. The force on the lander is $mv'(t)\vec{\imath}$ by Newton's second law. The forces $mg_0\vec{\imath}$ and $-mg_1\vec{\imath}$ add to $-mA\vec{\imath}$. Force competition $mv'(t)\vec{\imath} = -mA\vec{\imath}$ gives the velocity model

$$mv'(t) = -mA, \quad v(0) = v_0.$$

This quadrature-type equation is solved routinely to give

$$v(t) = -At + v_0, \quad x(t) = -A\frac{t^2}{2} + v_0 t.$$

The equation $v(T) = 0$ gives $T = v_0/A$ and $H = x(T) = v_0^2/(2A)$.

Numerical illustration. Let $v_0 = 1200$ miles per hour and $A = 30000$ miles per hour per hour. We compute values $T = 1/25$ hours = 2.4 minutes and $H = x(T) = 24$ miles. A `maple` answer check appears below.

```
v0:=1200; A:=30000;
X:=t->-A*t^2/2+v0*t;
T:=(v0/A): (T*60.0).'min',X(T).'miles'; # 2.4 min, 24 miles
A1:=A*2.54*12*5280/100/3600/3600; # mks units 3.725333334
v1:=v0*12*2.54*5280/100/3600;     # mks units 536.448
evalf(convert(X(T),units,miles,meters)); # 38624.256
```

The constant field model predicts that the retrorockets should be turned on 24 miles above the moon's surface with soft landing descent time of 2.4 minutes. It turns out that a different model predicts that 24 miles is too high, but only by a small amount. We investigate now this alternative model, based upon replacing the constant gravitational field by a variable field.

Variable Gravitational Field

The system of units will be the *mks* system. Assume the lunar lander is located at position P above the moon's surface. Define symbols:

$m =$ mass of the lander in kilograms,

$M = 7.35 \times 10^{22}$ kilograms is the mass of the moon,

$R = 1.74 \times 10^{6}$ meters is the mean radius of the moon,

$G = 6.6726 \times 10^{-11}$ is the universal gravitation constant, in *mks* units,

$H =$ height in meters of position P above the moon's surface,

$v_0 =$ lander velocity at P in meters per second,

$g_0 = GM/R^2 =$ constant acceleration due to the moon's gravity in meters per second per second,

$g_1 =$ constant retrorocket thrust deceleration in meters per second per second,

$A = g_1 - g_0 =$ effective retrorocket thrust deceleration in meters per second per second, constant field model,

$t =$ time in seconds,

$x(t) =$ distance in meters from the lander to position P,

$v(t) = x'(t) =$ velocity of the lander in meters per second.

The project is to find the height H above the moon's surface and the descent time T for a soft landing, using fixed retrorockets at time $t = 0$. The origin of coordinates will be P and $\vec{\imath}$ is directed from the lander to the moon. Then $x(t)\vec{\imath}$ is the lander position at time t. The initial conditions are $x(0) = 0$, $v(0) = v_0$. Let $g_0(t)$ denote the variable acceleration of the lander due to the moon's gravitational field. Newton's universal gravitation law applied to point masses representing the lander and the moon gives the expression

$$\text{Force} = mg_0(t)\vec{\imath} = \frac{GmM}{(R+H-x(t))^2}\vec{\imath}.$$

The force on the lander is $mx''(t)\vec{\imath}$ by Newton's second law. The force is also $mg_0(t)\vec{\imath} - mg_1\vec{\imath}$. Force competition gives the second order distance model

$$mx''(t) = -mg_1 + \frac{mMG}{(R+H-x(t))^2}, \quad x(0) = 0, \quad x'(0) = v_0.$$

The technique from the Jules Verne problem applies: multiply the differential equation by $x'(t)$ and integrate from $t = 0$ to the soft landing time $t = T$. The result:

$$\left.\frac{(x'(t))^2}{2}\right|_{t=0}^{t=T} = -g_1(x(T) - x(0)) + \left.\frac{GM}{R+H-x(t)}\right|_{t=0}^{t=T}.$$

Using the relations $x(0) = 0$, $x'(0) = v_0$, $x'(T) = 0$ and $x(T) = H$ gives a simplified implicit equation for H:

$$-\frac{v_0^2}{2} = -g_1 H + \frac{GM}{R} - \frac{GM}{R+H}.$$

Numerical illustration. Use $v_0 = 536.448$, $g_1 = 5.3452174$ to mimic the constant field example of initial velocity 1200 miles per hour and effective retrorocket thrust 30000 miles per hour per hour. A soft landing is possible from height $H = 23.7775$ miles with a descent time of $T = 2.385$ minutes. These results compare well with the constant field model, which had results of $H = 24$ miles and $T = 2.4$ minutes. Some `maple` details follow.

```
M:=7.35* 10^(22);R:=1.74* 10^6;G:=6.6726* 10^(-11);
v0_CFM:=1200: A_CFM:=30000: # Constant field model values
cf:=1*5280*12*2.54/100/3600: # miles/hour to meters/second
v0:=v0_CFM*cf; g0:=G*M/R^2: g1:=A_CFM*cf/3600+g0;
```

```
eq:= -(v0^2/2) + g1*H + G*M/(R+H) - G*M/R=0:
HH:=[solve(eq,H)][1];  # HH := 38266 meters
de:=diff(x(t),t,t) = -g1 + M*G/(R+HH-x(t))^2;
ic:= x(0)=0, D(x)(0)=v0;
with(DEtools):
DEplot(de,x(t),t=0..290,[[ic]]); # See the plot below
p:=dsolve({de,ic},x(t),numeric):
X:=t->evalf(rhs(p(t)[2])):
V:=t-> evalf(rhs(p(t)[3])):
TT1:=fsolve('V(t)'=0,t,100..800): TT:=TT1/60:
TT1.'seconds', TT.'minutes';
X(TT1).'meters', ((X(TT1)*100/2.54)/12/5280).'miles';
```



Figure 9. A `maple` plot used to determine the descent time $T = 2.385$ **minutes.**

Modeling

The field of the earth has been ignored in both models, which is largely justified because the universal gravitation law term for the lander and the earth is essentially zero for lander locations near the moon.

The field for the lander and the moon is not constant, and therefore it can be argued that conditions exist when assuming it is constant will produce invalid and obviously incorrect results.

Are there cases when the answers for the two models differ greatly? Yes, but the height H of retrorocket activation has to be large. This question is re-visited in the exercises.

Control problems. The descent problem for a lunar lander is a control problem in which the **controller** is the retrorocket plus the duration of time in which it is active. All we have done here is to decide that the descent should be controlled by retrorockets well in advance of 24 miles above the moon's surface. The methods used here can be applied to gain insight into the **bang-bang control problem** of turning on the retrorockets for n intervals of time of durations $\Delta t_1, \ldots, \Delta t_n$ to make an *almost* soft landing.

Primitive numerical methods. The predictions made here using the computer algebra system `maple` can be replaced by primitive RK4 methods and graphing. No practising scientist or engineer would do *only* that, however, because they want to be confident of the calculations and the results. The best idea is to use a **black box** of numerical and graphical methods which have little chance of failure, e.g., a computer algebra system or a numerical laboratory.

Exercises 4.7

Lunar Lander Constant Field. Find the retrorocket activation time T and the activation height $x(T)$. Assume the constant gravitational field model. Units are miles/hour and miles/hour per hour.

1. $v_0 = 1210$, $A = 30020$.

2. $v_0 = 1200$, $A = 30100$.

3. $v_0 = 1300$, $A = 32000$.

4. $v_0 = 1350$, $A = 32000$.

5. $v_0 = 1500$, $A = 45000$.

6. $v_0 = 1550$, $A = 45000$.

7. $v_0 = 1600$, $A = 53000$.

8. $v_0 = 1650$, $A = 53000$.

9. $v_0 = 1400$, $A = 40000$.

10. $v_0 = 1450$, $A = 40000$.

Lunar Lander Variable Field. Find the retrorocket activation time T and the activation height $x(T)$. Assume the variable gravitational field model and *mks* units.

11. $v_0 = 540.92$, $g_1 = 5.277$.

12. $v_0 = 536.45$, $g_1 = 5.288$.

13. $v_0 = 581.15$, $g_1 = 5.517$.

14. $v_0 = 603.504$, $g_1 = 5.5115$.

15. $v_0 = 625.86$, $g_1 = 5.59$.

16. $v_0 = 603.504$, $g_1 = 5.59$.

17. $v_0 = 581.15$, $g_1 = 5.59$.

18. $v_0 = 670.56$, $g_1 = 6.59$.

19. $v_0 = 670.56$, $g_1 = 6.83$.

20. $v_0 = 715.26$, $g_1 = 7.83$.

Distinguishing Models. The constant field model (**1**) (page 268) and the variable field model (**2**) (page 269) are verified here, by example, to be distinct. Find the retrorocket activation times T_1, T_2 and the activation heights $x_1(T_1)$, $x_2(T_2)$ for the two models (**1**), (**2**). Relations $A = g_1 - g_0$ and $g_0 = GM/R^2$ apply to compute g_1 for the variable field model.

21. $v_0 = 1200$ mph, $A = 10000$ mph/h. Answer: 72, 66.91 miles.

22. $v_0 = 1200$ mph, $A = 12000$ mph/h. Answer: 60, 56.9 miles.

23. $v_0 = 1300$ mph, $A = 10000$ mph/h. Answer: 84.5, 77.7 miles.

24. $v_0 = 1300$ mph, $A = 12000$ mph/h. Answer: 70.42, 66.26 miles.